

A close-up photograph of a hand with red-painted fingernails holding a thick, open book. The book's pages are fanned out, creating a sense of depth and movement. The background is a soft, teal-colored wall. The overall composition is artistic and focused on the act of reading or presenting a book.

Creative Coding

Implementing an ∞ -Exhibition

BEAT ROSSMY AND ALBRECHT SCHMIDT

EDITION 2019, CLASS AT LMU MUNICH

Creative Coding

Implementing an ∞ -Exhibition

BEAT ROSSMY AND ALBRECHT SCHMIDT

EDITION 2019

THIS BOOK IS WORK IN PROGRESS AND BASED ON A CLASS AT LMU MUNICH

Copyright © 2019 what license? should discuss thi...



Table of Contents

1	Programming and Creative Coding	7
1.1	Concept of this Book	7
1.2	The Goal: an ∞ -Exhibition	7
1.3	Target Group and Expected Skills	8
2	Aesthetics and Information	9
2.1	Programming a Random Mondrian	9
2.2	Random Creation and Aesthetics	10
2.3	Encoding Information in Art	11
2.4	Tasks - 1st lecture	12
3	Pixel based Art	14
3.1	Programming a "Pixel Converter"	14
3.2	Convert to: Pointillism	16
3.3	Convert to other styles	16
3.4	Tasks - 2nd lecture	16
4	Virtual Brushes	18
4.1	Programming a "virtual brush"	18
4.2	Implement: Finite Color	20

4.3	Image Brushes	20
4.4	Tasks - 3rd lecture	20
5	Lines	21
5.1	Frieder Nake	21
5.2	Implement: Specify Angles	23
5.3	What else to implement?	23
5.4	Tasks - 3rd lecture	23
6	Fractals and Recursive Structures	24
6.1	Nature	24
6.2	How can we change the code?	26
6.3	Tasks - 5rd lecture	26
7	Input and Output	27
7.1	Intro	27
7.2	Video	28
7.3	Audio	28
7.4	Tasks - 7th lecture	29



1. Programming and Creative Coding

1.1 Concept of this Book

The concept of the book is to get people excited for programming artistic pieces. In this book we teach basic concepts that make up building blocks for interactive and potentially artistic programs. The focus is to help the reader to create creative software and to learn how to program such software. The artistic part is up to the reader, but we hope to give good starting points, where readers can bring in their creativity.

In parts we suggest critical reflection on creative coding and programming art pieces. In the lecture at LMU this is done as discussions, that are not captured in the book. We recommend readers to show their ideas and programs to other and discuss and defend it. This aids reflection and learning, similar to what we do in the course.

Besides the programming concepts and the implementation techniques, we discuss different topics, including:

- How aesthetics can be programmed?
- Creativity vs. determined by algorithms?
- Creative information visualization vs. art?
- Evolutionary systems and creativity?
- ...

1.2 The Goal: an ∞ -Exhibition

In the course we create different interactive software programs. With the selection of the topics we try to cover many different programming concepts and implementation techniques.

All the programs are designed to create changing new outputs, either after some time or based on user interaction. All programs together can then be set-up as an exhibition that is ever changing, which we call an ∞ -Exhibition. If the screens with the software are set-up in a circle, where screens face outwards, a visitor to exhibition can walk around and will always see new exhibits.

1.3 Target Group and Expected Skills

The course and book is designed for readers studying at the intersection of art, media, and technology. We expect that readers have an interest in interactive art and that they want to improve their abilities to create interactive installations.

We expect that readers have basic programming skills in processing. If readers have no programming skills at all, we suggest to do an introductory tutorial (e.g. <https://processing.org/tutorials/>). For readers with programming experience in another languages, we expect that they can follow the course and read up on processing details as they go along.



2. Aesthetics and Information

2.1 Programming a Random Mondrian

Implement a program that creates a drawing that is inspired by Mondrian.




Figure 2.1: Example of a Mondrian Painting.

The program should create a new picture every 120 seconds or when the user presses a button on the keyboard.

Useful Commands

```
1 void keyPressed ();  
2 random ();  
3 randomSeed ();  
4 second ();  
5 minute ();  
6 hour ();
```

The following listing is a starting point for your project.

```
1 size(400,400); // drawing canvas, coordinate systems
2 background(255); // background is white
3
4 fill(255, 0, 0); // color is red
5 rect(50, 100, 250, 100);
6
7 fill(0, 255, 0); // color is green
8 rect(250, 250, 100, 100);
9
10 fill(0, 0, 255); // color is blue
11 rect(50, 320, 150, 10);
12
13 fill(255, 255, 0); // color is yellow
14 rect(50, 340, 150, 6);
15
16 fill(255, 255, 255); // color is white
17 rect(50, 360, 150, 4);
18
19 fill(0, 0, 0); // color is black
20 rect(50, 380, 150, 3);
```




Figure 2.2: Resulting Image from the example code.

See https://processing.org/reference/random_.html for examples how to use the random function.

2.2 Random Creation and Aesthetics

Experiment with the parameters in the random creation. Explore the impact of parameters such as the size, dimension, line width, colors, and number of element on the aesthetics of the resulting picture.

Why are some pictures are aesthetically more pleasing than others? How can you change your algorithm to always create aesthetic drawings, while still having a great variety?

2.3 Encoding Information in Art

People have discussed the idea of informative art, see [Fer07; HS03; RSH00]. Can this utilitarian approach to creating visual displays based on know artworks be considered art?

Create a display that shows the time. Take a well know artwork as inspiration and encode the time in your visualization (a 15 minutes resolution is sufficient). To a casual observer it should not be apparent that the visualization shows the time.

What are the difficulties in designing such a visual display?

The following listing shows a basic clock. You can use this as a starting point for your own project.

```
1 // Variablen definieren - ausserhalb von setup() und draw()
2 // da sie in beiden Funktionen verwendet werden
3 int stunde, minute, sekunde;
4
5 void setup() {
6 // Fenstergroesse festlegen
7   size(260, 100);
8
9   // Variablen mit 0 initialisieren
10  stunde = 0;
11  minute = 0;
12  sekunde = 0;
13
14  // die Funktion draw() wird 2x pro Sekunde ausgefuehrt
15  frameRate(2);
16 }
17
18 void draw() {
19   // loesche den Hintergrund: setze ihn schwarz20background(0);
20   // setze Zeichenfarbe fuer Text und Recheck: weiss
21   fill(255, 255, 255);
22
23   // speichere in den Variablen die aktuelle Zeit
24   stunde = hour();
25   minute = minute();
26   sekunde = second();
27
28   // Kontrollausgabe der Zeit in der Console - kann spaeter
29   // in processing kann man mit "+" einen String aus Buchstaben und
30   // Zahlenbauen
31   println(stunde + ":" + minute + ":" + sekunde);
32
33   // zeige die Zeit im Fenster an
34   textSize(32);
35   text(stunde + ":" + minute + ":" + sekunde, 10, 30);
36
37   // zeichne einen Sekundenbalken
38   // Sekunde hat einen wert von 0 bis 59
39   // Sekunde multipliziert mit 4 hat einen Wert zwischen 0 und 236
40   rect(10, 40, sekunde*4, 20);
41 }
```

There is another example of an analog clock that may be helpful: <https://processing.org/examples/clock.html>

2.4 Tasks - 1st lecture

In the 1st session we try to mimic paintings of the famous dutch painter Piet Mondrian. He is famous for his abstract paintings that are reduced to a minimal palette of colors and geometric structures. His systematic approach invites to be recreated with algorithms.

Preparation

If Processing isn't already installed on your computer, go to <https://processing.org/download/> download the program and install it on your machine. Open some example sketches and familiarize yourself with the Processing IDE. If you feel comfortable, proceed with the following task.

Breakout Session

During the following task we try to approach the generation of random Mondrianesque images in a systematic way. Step by step we go from the analysis of real and fake Mondrians, over theoretical concepts for the description of Mondrianesque structures, to working implementations of the algorithm.

- Search on the internet for pictures of the painter Mondrian (see Figure 2.3). What are the connecting commonalities? Can you spot artworks not painted by Mondrian? Try to figure out the "recipe" that makes a distinguishable Mondrian.
- Let us discuss your approach to generate Mondrianesque images with the help of an algorithm. Try to sketch such structures and formalize them first.
- Try to implement one of the approaches we discussed in the plenum. Does it lead to the expected result? Does it work as simply as expected or have there been unexpected complications?
- If your algorithm is working, try to extend it to add your own style to the generated "Mondrians". Tweak parameters, change the color palette or even introduce more randomness.




Figure 2.3: Artworks painted by Mondrian and Mondrian-inspired artworks.

Discussion

We just created algorithms that mimic the style of a famous artist. Let us discuss your impressions and thoughts along the way. As a starting point think about the following questions:

- Can you call the product of these algorithms "art"?
- Can art in general be scripted? (Pollock, Warhol, ...)
- Does the used technology influence such a differentiation? (Style transfer performed with neural networks vs classic "conditional" algorithms)

Homework

1. **Reading:** Read the following papers [Fer07; HS03; RSH00] and write a short (300-500 words) essay on Informative Art. Your essay should contain an explanation what it is, why people would want to do it, pros and cons.
2. **Implementation:** Create a display that shows the time. Take Mondrian algorithm and encode the time in your visualization (a 15 minutes resolution is sufficient).
3. Collect feedback on your implementation. Record 3 videos from people seeing your implementation (15 seconds each).



3. Pixel based Art

3.1 Programming a "Pixel Converter"

Implement a program that takes an image as an input, analyzes the content and transfers it to a new output design. As an example look at pictures from Pointillism, ASCII art, etc..



Figure 3.1: The shown examples demonstrate the translation of a color information to a interpreted output.

The program should take a predefined image from your file system or a snapshot over the integrated webcam of your computer.

Useful Commands

```
1 PVector vec;  
2 loadImage("testfile.jpg");  
3 IntList list;  
4 color c;  
5 red(c);  
6 green(c);  
7 blue(c);
```

The following code is a starting point for your project.

```
1 PImage source;
2 int area = 25;
3
4 void setup () {
5
6   size(900,600);
7   source = loadImage("test.jpg");
8   noStroke();
9
10  for (int y=0; y<source.height; y+=area) {
11    for (int x=0; x<source.width; x+=area) {
12      fill(getAverageColor(x,y,area,area,source));
13      rect(x,y,area,area);
14    }
15  }
16 }
17
18 color getAverageColor ( int x, int y, int w, int h, PImage img) {
19   float r = 0;
20   float g = 0;
21   float b = 0;
22
23   for (int i=0; i<w*h; i++) {
24     int index = x+(i%w)+y*img.width+(img.width*(i/w));
25     r += red(img.pixels[index]);
26     g += green(img.pixels[index]);
27     b += blue(img.pixels[index]);
28   }
29
30   return color(r/(w*h),g/(w*h),b/(w*h));
31 }
```




Figure 3.2: Resulting Image from the example code.

3.2 Convert to: Pointillism

Try to extend the given code to approximate the style of a pointillistic artwork:

- Draw many circles instead of one big square.
- Randomize the position of these circles.
- Randomize color values.

What effects can you observe that interfere with the pointillistic style? Try to make your algorithm generate more vivid outcomes. If you look closely at the pointillistic image shown in the beginning of the chapter, one can observe that the painter uses small colors dots of primary colors that are mixed in our brain to multiple shades of the color palette. Can we imitate this effect?

3.3 Convert to other styles

In the first example code a function was given that translates a given area of an image to the average color of this area. What other informations are contained in such areas?

- color spread
- color palette
- ...

Try to define functions that take an image, coordinates and dimensions of the subarea to be observed and extract these informations.

```

1 color getAverageColor ( int x, int y, int w, int h, PImage img) {
2   float r = 0;
3   float g = 0;
4   float b = 0;
5
6   for (int i=0; i<w*h; i++) {
7     int index = x+(i%w)+y*img.width+(img.width*(i/w));
8     r += red(img.pixels[index]);
9     g += green(img.pixels[index]);
10    b += blue(img.pixels[index]);
11  }
12
13  return color(r/(w*h),g/(w*h),b/(w*h));
14 }
15
16 ArrayList<Integer> getColors ( int x, int y, int w, int h, PImage img) {
17   ArrayList<Integer> colors = new ArrayList<Integer>();
18
19   for (int i=0; i<w*h; i++) {
20     int index = x+(i%w)+y*img.width+(img.width*(i/w));
21     color c = img.pixels[index];
22     if (!colors.contains(c)) colors.add(c);
23   }
24
25   return colors;
26 }
27
28 ...

```

3.4 Tasks - 2nd lecture

In the 2nd session we try to process image information and translate these into new "styles". As a source of inspiration we discuss the genre of pointillism as well as pictures from the ASCII art

scene. Further, we try to go beyond these two styles and approach new ideas and concepts.

Breakout Session

During the following task we try to think of our algorithms as transfer functions. With an image given we take the contained information (colors of pixels) and transfer it into: colored shapes, ASCII characters, ...

In this model the image is the input, which is then analyzed on a subarea level. The contained information is transformed by a small instruction set and thereby applied to the whole image.

- Search on the internet for pointillistic images and for examples of ASCII art.
- Discuss the different concepts how colors, shapes and more are abstracted.
- Start to implement your analyzed concept based on the previously given example code.
- Where are the weak points of the algorithm? Can you spot its algorithmic nature? How can these problems be addressed and solved?

Discussion

What are the most interesting aspects of this approach? Let us discuss your impressions and thoughts along the way. As a starting point think about the following questions:

- Can you call the product of these algorithms "art"?
- For now, we have reduced information (subareas), can we also extend to a higher richness of detail?
- Could you imagine to use such a technique in your own work?

Homework

1. **Reading:** Read the following papers <http://www.cse.cuhk.edu.hk/~ttwong/papers/asciiart/asciiart.pdf> to get an idea about the challenges and approaches to ASCII art.
2. **Implementation:** Look into the following webpage <http://paulbourke.net/dataformats/asciiart/> and adopt these rules for ASCII art to our algorithm. Implement a "Convert to: ASCII" sketch.
3. Think of the following: For the moment our ASCII sketch produces random character sequences based on the color value of the source pixels. Can we use words instead of random sequences? What do you have to do to take as an additional input something like Shakespeare's collected work and search in there for matching "color/character" sequences? Try an implementation.



4. Virtual Brushes

4.1 Programming a "virtual brush"

In this section we will look into brushes. Brushes are a major part of the process of painting. A brush stroke can tell you a lot about the process, the intention and the concept the artist used during the creation process.



Figure 4.1: Different brush types are used to create different strokes on the canvas.

Our implementation can consider the following attributes: direction, speed, color, brush type, ...

Useful Commands

```
1 PVector vec;  
2 pixels[];  
3 red(c);  
4 green(c);  
5 blue(c);
```

The following code is a starting point for your project.



Figure 4.2: Different painting techniques tell many things about the process and can create different feelings.

```

1 PVector lastMouse, currentMouse;
2 color brushColor;
3
4 void setup () {
5   size(800,800);
6   lastMouse = new PVector(0,0);
7   currentMouse = new PVector(0,0);
8   brushColor = color(255,0,0);
9   background(255);
10  noStroke();
11 }
12
13 void draw () {
14   lastMouse = currentMouse.copy();
15   currentMouse = new PVector(mouseX,mouseY);
16   brushStroke(brushColor);
17 }
18
19 void brushStroke (color c) {
20   if (!mousePressed) return;
21
22   if (lastMouse!=currentMouse && freshness>0) freshness--;
23
24   PVector direction = PVector.sub(currentMouse,lastMouse);
25
26   fill(brushColor);
27   ellipse(lastMouse.x,lastMouse.y,50,50);
28 }
29
30 void keyPressed () {
31   if (key == '1') brushColor = color(255,0,0);
32   if (key == '2') brushColor = color(0,255,0);
33   if (key == '3') brushColor = color(0,0,255);
34 }

```

4.2 Implement: Finite Color

Try to extend the given code to simulate the effect of the brush running out of color.

- Add a variable that is decremented as long as the brush performs a movement.
- Can you pick up color that is already on the canvas?

How can we improve the brush stroke? Can we use the speed of the mouse to control parameters? You can already find the speed of the mouse indirectly in the code. Where?

4.3 Image Brushes

Can we use images to create more realistic brushes? What do we need for that?

- How to convert an image into a reusable and adaptable brush?
- How to react to different directions?
- How to react to time based changes of the brush stroke?

Discuss and describe how you would approach this problem in your group.

4.4 Tasks - 3rd lecture

In the 3rd session we want to discuss how we can implement expressive and vivid brush strokes in processing using the directional vector.

Breakout Session

During the following task we take a closer look at different kinds of brushes and strokes.

- Search on the internet for examples of different artists and styles.
- Discuss the different factors leading to these outcomes.
- what has the biggest influence?
- How can we translate this into an algorithm?
- Expand the example code with some of your ideas. Try to integrate time-based interactions.

Homework

1. **Implementation:** Implement based on the example code and your experimentation during the session a more elaborated version of this painting sketch.
2. **Implementation:** Integrate a more detailed way to choose colors, brush sizes, ...



5. Lines

5.1 Frieder Nake

In this section we will look into the algorithmic art of Frieder Nake. Nake is famous for his first explorations of the "Graphomat" (today we call such machines plotters). He decided to experiment with this new type of machines in an artistic way during his task to write software for this new hardware which was targeted at architects and engineers to speed up and ease the drawing process. Instead of typical calibration tests he wrote algorithms that "created" artistic and randomly generated drawings.




Figure 5.1: Rule sets can specify angles minimum and maximum distances,

We will try to implement to implement an algorithm that generates similar aesthetics. Further, it would be interesting to "watch" the algorithm draw. How can we follow along with the "plotter" during the "creation" process?

Useful Commands

```

1 PVector vec;
2 PVector.random2D();
3 PVector.add(vec1,vec2);
4 PVector.sub(vec1,vec2);
5 PVector.mult(vec1,f);
6 random();
7 constrain();

```




Figure 5.2: Cage's work is manifold and nevertheless beautiful in its simplicity.

The following code is a starting point for your project.

```

1 PVector plotterHead,last;
2 PVector direction;
3
4 void setup () {
5   size(800,800);
6   smooth();
7   background(255,250,240);
8   stroke(0,100);
9   strokeWeight(4);
10
11  plotterHead = new PVector(random(800),random(800));
12  last = plotterHead.copy();
13  direction = PVector.random2D();
14 }
15
16 void draw () {
17  plotterHead = PVector.add(plotterHead,direction);
18  line(plotterHead.x,plotterHead.y,last.x,last.y);
19
20  last = plotterHead.copy();
21  if (random(100)<=1 || (plotterHead.x>=width || plotterHead.x<=0 ||
22     plotterHead.y>=height || plotterHead.y<=0)) {
23    direction = PVector.random2D();
24  }

```

5.2 Implement: Specify Angles

Try to give the algorithm a "rule set" that specifies the angles to be used. Should there be just 90 degree angles or random angles or ...?

- Just pick specific angles instead of random ones.
- Think of using an array that contains the valid angles?

How can we apply more complex rules such as: 90 degree lines are longer than 60 degree ones.

5.3 What else to implement?

If we compare our outcome to the found drawings of Nake's algorithms where are the differences?

- Try to find and point out as many differences as occur to you.
- Do you think every drawing of the algorithm is an artwork?
- Did the artist pick the best ones?

Discuss the given tasks and questions in your group.

5.4 Tasks - 3rd lecture

In the 4th session we want to discuss how we can implement simple abstract and line-based algorithmic art.

Breakout Session

During the following task we take a closer look at different kinds of brushes and strokes.

- Search on the internet for examples of Nake's work.
- Discuss the different parameters he used over the course of his different cycles.
- What role do randomness and order play in his work?
- Is the beauty just in the aesthetic of the drawings or as well in the simplicity of the logic behind?
- Expand the example code with some of your ideas. Try to find the most elegant solution for your specific problem.

Homework

1. **Reading:** Read through the following web document http://dreher.netzliteratur.net/4_Medienkunst_Kybernetik.html and this paper written by Frieder Nake [Nak05] <http://citeseerx.ist.psu.edu/viewdoc/download?doi=10.1.1.470.2597&rep=rep1&type=pdf>
2. **Implementation:** Implement two of Nake's other algorithms (as shown in Figure 5.2 or found online).



6. Fractals and Recursive Structures

6.1 Nature

In this section we will look into common structures known from nature called "Fractals". Fractals are structures that reoccur in them self. Common examples would be: mountains, snow flakes, trees. All these structures show the same features from a distant point of view as well as under close examination. A tree is just a single branch from which other branches emanate, from which smaller branches emanate from which ... In nature these structures come to an logical end. The last branch emanates into leaves which mark the end of the structure, but on a mathematical level these structures can go into infinity. The most prominent example for mathematical fractals is the Mandelbrot Fractal.




Figure 6.1: Even every day objects as broccoli reveal fractal qualities.

We will try to implement a fractal in a generic manner and from there explore the impact of different rule sets, randomness or geometrical elements (lines, circles, ...) on the visual outcome.

Useful Commands

```
1 PVector vec;
```

```

2 PVector.sub(vec1,vec2);
3 vec.rotate(f);
4 ArrayList<Fractal> children;

```

The following code is a starting point for your project.

```

1 class Fractal {
2   ArrayList<Fractal> children;
3   Fractal parent;
4   int level = 0;
5   int max = 10;
6
7   PVector start, end;
8
9   public Fractal (float sx, float sy, float ex, float ey) {
10    parent = null;
11    level = 0;
12    children = new ArrayList<Fractal>();
13    start = new PVector (sx,sy);
14    end = new PVector (ex,ey);
15
16    PVector dir = PVector.sub(end,start);
17
18    children.add(new Fractal(this,dir));
19    children.add(new Fractal(this,dir.rotate(-0.2)));
20    children.add(new Fractal(this,dir.rotate(0.4)));
21  }
22
23  public Fractal (Fractal parent, PVector d) {
24    this.parent = parent;
25    level = parent.level+1;
26    children = new ArrayList<Fractal>();
27
28    start = parent.end;
29    end = PVector.add(start,d);
30
31    if (exitCondition()) return;
32    PVector dir = PVector.sub(end,start);
33
34    children.add(new Fractal(this,dir));
35    children.add(new Fractal(this,dir.rotate(-0.2)));
36    children.add(new Fractal(this,dir.rotate(0.4)));
37  }
38
39  boolean exitCondition () {
40    return level>10;
41  }
42
43  void draw () {
44    line(start.x,start.y,end.x,end.y);
45    for (int c=0; c<children.size(); c++) children.get(c).draw();
46  }
47 }
48
49 void setup () {
50   size(800,800);
51   Fractal f;
52   f = new Fractal(400,800, 400, 750);
53   stroke(0,10);
54   f.draw();
55 }

```

6.2 How can we change the code?

Try to change parameters of the code.

- Choose different angles.
- Introduce randomness.
- Branches getting shorter.
- ...

6.3 Tasks - 5rd lecture

In the 5rd session we want to discuss how we can use fractal structures to recreate structures known from nature.

Breakout Session

During the following task we take a closer look into star systems.

- Compare these systems to the fractals we have seen so far. Do the same/similar rules apply?
- What features do we have to add to our code to improve the visual effect.
- what role do randomness and order play here?

Homework

1. **Reading:** Read through the following web document http://dreher.netzliteratur.net/4_Medienkunst_Kybernetik.html and this paper written by Frieder Nake [Nak05] <http://citeseerx.ist.psu.edu/viewdoc/download?doi=10.1.1.470.2597&rep=rep1&type=pdf>
2. **Implementation:** Implement two of Nake's other algorithms (as shown in Figure 5.2 or found online).



7. Input and Output

7.1 Intro

Most of the time we want create sketches that not only generate themselves but also can react to or even depend on the interaction with the viewer. Such interactions can start with typical PC peripheral devices such as the mouse and the keyboard but can also go way beyond these. In this section we will have a short look into the options that are easily integrable into processing sketches. Therefore we will have a look into: live video stream, audio stream analysis, custom sensors with Arduino.



Figure 7.1: Typical peripheral devices can be used to enhance the interaction with generative sketches. But also custom hardware can be used and integrated for this purpose.

Useful Commands

```
1 int x = mouseX;  
2 int y = mouseY;  
3 char c = key;  
4 boolean b = keyPressed;  
5  
6 keyPressed()  
7 keyReleased()
```

7.2 Video

```
1 // from: https://processing.org/reference/libraries/video/Capture.html
2
3 import processing.video.*;
4
5 Capture cam;
6
7 void setup() {
8     size(640, 480);
9
10    String[] cameras = Capture.list();
11
12    if (cameras.length == 0) {
13        println("There_are_no_cameras_available_for_capture.");
14        exit();
15    } else {
16        println("Available_cameras:");
17        for (int i = 0; i < cameras.length; i++) {
18            println(cameras[i]);
19        }
20
21        // The camera can be initialized directly using an
22        // element from the array returned by list():
23        cam = new Capture(this, cameras[0]);
24        cam.start();
25    }
26 }
27
28 void draw() {
29     if (cam.available() == true) {
30         cam.read();
31     }
32     image(cam, 0, 0);
33     // The following does the same, and is faster when just drawing the
34     // image
35     // without any additional resizing, transformations, or tint.
36     //set(0, 0, cam);
37 }
```

7.3 Audio

```
1 // from: https://processing.org/reference/libraries/video/Capture.html
2
3 import processing.sound.*;
4
5 FFT fft;
6 AudioIn in;
7 int bands = 512;
8 float[] spectrum = new float[bands];
9
10 void setup() {
11     size(512, 360);
12     background(255);
13
14     // Create an Input stream which is routed into the Amplitude analyzer
15     fft = new FFT(this, bands);
16     in = new AudioIn(this, 0);
17
18     // start the Audio Input
```

```
19  in.start();
20
21  // patch the AudioIn
22  fft.input(in);
23  }
24
25  void draw() {
26    background(255);
27    fft.analyze(spectrum);
28
29    for(int i = 0; i < bands; i++){
30      // The result of the FFT is normalized
31      // draw the line for frequency band i scaling it up by 5 to get more
        amplitude.
32      line( i, height, i, height - spectrum[i]*height*5 );
33    }
34  }
```

7.4 Tasks - 7th lecture

Work on your final projects, try to sketch the rough structure of your processing sketch. What are the major challenges of your idea? What do you need as additional input?

Homework

As the homework prepare and document one your favorite sketches you implemented over the course of the semester. We will use these sketches as illustrative material for the next year's course. Please do the following:

1. Pick your favorite sketch you implemented so far,
2. review your code and try to simplify it as far as possible (use clear variable names, use methods to avoid redundancy, ...),
3. go through your code and add comments to make the sketch comprehensible, add a description of your sketch in the beginning of the sketch,
4. make a high quality rendering of the outcome (.png, .jpg, ...).

Collect all the materials and hand them in as a .zip file.



Bibliographie

- [Fer07] Alois Ferscha. “Informative art display metaphors”. In: *International Conference on Universal Access in Human-Computer Interaction*. Springer. 2007, pages 82–92 (cited on pages 11, 13).
- [HS03] Lars Erik Holmquist and Tobias Skog. “Informative art: information visualization in everyday environments”. In: *Proceedings of the 1st international conference on Computer graphics and interactive techniques in Australasia and South East Asia*. ACM. 2003, pages 229–235 (cited on pages 11, 13).
- [Nak05] Frieder Nake. “Computer art: a personal recollection”. In: *Proceedings of the 5th conference on Creativity & cognition*. ACM. 2005, pages 54–62 (cited on pages 23, 26).
- [RSH00] Johan Redström, Tobias Skog, and Lars Hallnäs. “Informative art: using amplified artworks as information displays”. In: *Proceedings of DARE 2000 on Designing augmented reality environments*. ACM. 2000, pages 103–114 (cited on pages 11, 13).